
1 ZÁKLADY

Co je Heidenhain
Kdo je HEIDENHAIN, jaké řady řízení vyrábí a čím se popisný dialog liší od G kódu.

HEIDENHAIN je německý výrobce řídicích systémů a odměřovacích zařízení. V dílně je známý
hlavně popisným dialogem, který se anglicky jmenuje Klartext. Systém se tě ptá a ty odpovídáš.
Tento manuál ukazuje programy i pro iTNC 530, pro který je napsaný program k výkresu Praktická
MZ_2026.

Řada Stroj Styl programování

iTNC 530, TNC 620, TNC 640 frézky a obráběcí centra popisný dialog, u některých
i DIN/ISO

MANUALplus 620, CNC PILOT 640 soustruhy cykly, režim Naučit, DIN

Obr. 1.1  Řídicí panel iTNC 530 pro frézky a řídicí panel CNC PILOT 640 pro soustruhy, oba z titulních stran příruček.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
1; Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 1

PAMATUJ

Mezi řadami jsou rozdíly. Frézovací část vychází z příručky TNC 620 a příručky cyklů iTNC 530,
soustružnická z CNC PILOT 640 a MANUALplus 620. Názvy tlačítek a opce se mohou lišit podle
stroje a verze softwaru.

V čem se liší popisný dialog od G kódu
G kód Popisný dialog

Přímka G1 X10 Y20 F100 L X+10 Y+20 F100

Rychloposuv G0 X10 L X+10 FMAX

Oblouk G2 X20 Y10 R5 CR X+20 Y+10 R+5 DR-

Korekce G41, G42, G40 RL, RR, R0

Podprogram podle systému LBL a CALL LBL

Cyklus podle systému CYCL DEF a CYCL CALL

Každý blok má číslo a na obrazovce se píše pomocí dialogu: stiskneš klávesu dráhy a řízení se
postupně ptá na souřadnice, korekci a posuv. Díky tomu se při psaní těžko zapomene, co je potřeba.

Obr. 1.2  Dialog na obrazovce iTNC 530: modrý řádek se ptá na další údaj, zde stoupání závitu (Pitch?), vedle je
pomocný obrázek k cyklu.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
50

2 ZÁKLADY

Popisný dialog podrobně
Kostra programu, dráhy, korekce, podprogramy a parametry Q v popisném dialogu.

Kostra programu

0  BEGIN PGM CEP MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 1 Z S2000
4  L Z+100 R0 FMAX
5  L X+10 Y+10 R0 FMAX M3
...
20 L Z+100 R0 FMAX M2
21 END PGM CEP MM

Obr. 2.1  Krátký program na obrazovce iTNC 530:
začátek a konec programu, polotovar, volání nástroje
a odjetí.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 54

Blok Význam

BEGIN PGM, END PGM začátek a konec, název a jednotky

BLK FORM 0.1, 0.2 polotovar pro grafiku, spodní a horní roh, osa nástroje

TOOL CALL číslo nástroje, osa, otáčky. Volitelně posuv a přídavky DL, DR

L přímka, X, Y, Z, korekce, posuv, M funkce

M2 nebo M30 konec programu

2  Popisný dialog podrobně

Dráhy
Zápis Význam

L X+50 Y+20 RL F200 přímka s korekcí, posuv 200

CC X+25 Y+25 střed kružnice, platí pro následující C

C X+45 Y+25 DR- oblouk kolem CC do koncového bodu, DR- ve směru hodin

CR X+70 Y+95 R+30 DR- oblouk daným poloměrem, R+ do 180°, R- nad 180°

CT X+30 Y+30 oblouk tečně navazující na předchozí dráhu

RND R5 zaoblení rohu mezi dvěma dráhami

CHF 10 sražení rohu

APPR LT, APPR LCT nájezd na obrys přímkou nebo tečným obloukem

DEP LT, DEP LCT odjezd z obrysu

Obr. 2.2  Vlevo oblouk kolem středu CC od bodu S do bodu E, vpravo oblouky daným poloměrem R se směry DR+
a DR-.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s.
197 a 198

2  Popisný dialog podrobně

Obr. 2.3  Vlevo sražení rohu délkou 12, vpravo zaoblení rohu R5.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s.
194 a 195

Obr. 2.4  Vlevo nájezd na obrys z bodu PS přes PH do PA, vpravo odjezd z bodu PE do PN. Před nájezdem a po
odjezdu platí R0.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s.
187 a 190

POZOR

Nájezd a odjezd s korekcí patří mimo materiál. Mezi bloky s RL a RR musí být blok R0.

2  Popisný dialog podrobně

Korekce a směr obrábění
Zápis Nástroj Kolem dílu po směru hodin

RL vlevo od dráhy sousledné frézování

RR vpravo od dráhy nesousledné frézování

R0 střed na dráze bez korekce

Obr. 2.5  Korekce rádiusu: vlevo nástroj vlevo od dráhy (RL), vpravo vpravo od dráhy (RR).
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s.
162

Podprogramy a opakování
Zápis K čemu

LBL 1 ... LBL 0 podprogram, který se volá. Patří za konec hlavní části

CALL LBL 1 zavolá podprogram jednou

CALL LBL 1 REP 3 část programu od LBL 1 zopakuje třikrát, celkem tedy proběhne
čtyřikrát

CALL PGM NAZEV spustí jiný program. Po jeho konci řízení pokračuje dalším
blokem

FN 9: IF +0 EQU +0 GOTO LBL 99 nepodmíněný skok, hodí se pro přeskočení podprogramů

2  Popisný dialog podrobně

POZOR

Volaný program nesmí obsahovat M2 ani M30. Když má ve své části podprogramy LBL, přeskoč je
skokem FN 9 na LBL 99 před END PGM.

Parametry Q
Místo čísel se dají použít proměnné Q. Výpočty se píšou funkcemi FN. Ve školních úlohách se s Q
nejčastěji setkáš jako s parametry cyklů: Q201 je třeba hloubka.

FN 0: Q1 = +5
FN 1: Q2 = +Q1 + +3
L Z-Q1 F100

3 FRÉZKA: TNC A ITNC 530

Heidenhain TNC
Provozní režimy, vztažný bod, nástroje a první programy na řízení TNC.

Frézky s řízením Heidenhain typicky používají systémy TNC, například TNC 620 a TNC 640. Tato
kapitola vychází z české příručky TNC 620, princip je u příbuzných systémů stejný. Hlavní jazyk je 
popisný dialog (Klartext): systém se ptá a ty odpovídáš, nemusíš znát G kódy.

PAMATUJ

Některé funkce jsou opce. Cykly 251 až 257, grafická simulace (opce 20) nebo dotyková sonda (opce
17) nemusí být na každém stroji. Tlačítka a pořadí referencí popisuje i příručka ke stroji.

3  Heidenhain TNC

Provozní režimy

Obr. 3.1  Strojní a programovací režimy se volí vlastními klávesami.

Režim K čemu slouží

Ruční provoz Pojíždění osami, krokové polohování, nastavení vztažného bodu.

Ruční kolečko Pojíždění elektronickým kolečkem.

Polohování s ručním zadáním Krátký program pro jednoduchou práci, ukládá se do $MDI. Nelze
v něm RL a RR.

Programování Psaní programů, cykly, programovací grafika.

Testování programu Simulace bez pohybu, hledá chyby a porušení pracovního prostoru.

Po bloku a plynule Provádění programu. Po bloku se každý blok spouští klávesou
NC-START.

Obr. 3.2  Obrazovky řízení iTNC 530: Ruční provoz (Manual operation) a Polohování s ručním zadáním (Positioning with
manl.data input).
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s. 40

Obr. 3.3  Obrazovky řízení iTNC 530: Programování (Programming and editing) s programovací grafikou a Testování
programu (Test run).
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s. 41

Zapnutí a reference
1. Zapni napájení. Systém nastartuje a ohlásí výpadek napětí, hlášení vymaž.

2. Zapni řídicí napětí, systém zkontroluje nouzové vypnutí.

3. Přejeď referenční body: v předvoleném pořadí NC-STARTem pro každou osu, nebo směrovými
klávesami. S absolutním odměřováním to odpadá.

4. Jen na programování a testování reference není potřeba, přejet se dají později softklávesou.

Ruční provoz: manuální práce
■ Směrová klávesa pojíždí, dokud ji držíš.

■ Směrová klávesa + NC-START pojíždí trvale, zastaví NC-STOP.

■ Krokové polohování: softklávesa VELIKOST KROKU na ZAP, zadej přísuv do 10 mm, každý
stisk je jeden krok.

■ Posuv se mění softklávesou F, otáčky a M funkce v dialogu S, F, M.

3  Heidenhain TNC

Obr. 3.4  Klávesnice řízení iTNC 530 a přenosné elektronické ruční kolečko HR 410.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s. 39
a 48

Polohování s ručním zadáním
Na jednoduché ofrézování nebo vrtání stačí pár řádků bez ukládání programu:

0 BEGIN PGM $MDI MM
1 TOOL CALL 1 Z S2000       ; nástroj 1, osa Z, otáčky
2 L Z+200 R0 FMAX           ; odjetí
3 L X+50 Y+50 R0 FMAX M3    ; nad díru, zapnout vřeteno
4 CYCL DEF 200 VRTÁNÍ       ; parametry viz dále
5 CYCL CALL                 ; provést cyklus
6 END PGM $MDI MM

Nastavení vztažného bodu
1. Upni a vyrovnej obrobek. Bez 3D sondy musí být rovnoběžně s osami.

2. Vlož nulový nástroj se známým rádiusem, zvol Ruční provoz.

3. Opatrně najeď, až se nástroj dotkne obrobku. Když se povrch nesmí poškodit, polož plech známé
tloušťky d.

4. Zvol osu a nastav indikaci. V ose nástroje zadej 0, případně d. V rovině obrábění počítej
s rádiusem nástroje.

5. Stejně nastav další osy.

TIP

Na roh obrobku v X: fréza Ø10 se dotkne levé strany, střed je 5 mm vlevo od nuly. Zadáš X = -5, ne
0.

3  Heidenhain TNC

Nástroje
Data nástrojů jsou v tabulce TOOL.T: délka L, rádius R a přídavky DL a DR. Softklávesu EDIT zapni,
šipkami vyber nástroj a hodnotu, klávesou END odejdeš. V programu se nástroj volá blokem TOOL
CALL 5 Z S4500, tedy číslo, osa a otáčky. Korekce rádiusu počítá s R a s přídavky z tabulky
i z TOOL CALL.

Program v popisném dialogu
Nový program založíš klávesou PGM MGT, jméno s příponou .H a jednotky MM. První a poslední
blok vytvoří systém sám. Doporučená stavba podle příručky:

1. Vyvolání nástroje a odjetí.

2. Předpolohování v rovině poblíž startu obrysu.

3. Předpolohování nad obrobek nebo rovnou do hloubky, vřeteno a chlazení.

4. Najetí na obrys, obrábění, opuštění obrysu.

5. Odjetí a konec programu.

Blok Význam

BLK FORM 0.1 / 0.2 polotovar pro simulaci, bod MIN a bod MAX

TOOL CALL 5 Z S4500 nástroj, osa, otáčky

L X+10 Y+20 R0 FMAX přímka rychloposuvem, bez korekce

L Y+95 RL F300 přímka s korekcí vlevo

CHF 10 zkosení v rohu, délka 10

RND R10 zaoblení v rohu, R10

CR X+70 Y+95 R+30 DR- oblouk daným poloměrem, DR- ve směru hodin

CT X+40 Y+5 tečný oblouk, poloměr si systém dopočítá

APPR LT, APPR LCT nájezd na obrys přímkou nebo obloukem tečně

DEP LT, DEP CT opuštění obrysu

CYCL DEF, CYCL CALL, M99 definice a provedení cyklu

M2 konec programu, M13 vřeteno a chlazení

3  Heidenhain TNC

Příklad: obrys se zkosením
Polotovar 100 x 100 x 20, nula na horní ploše. Převzato z příručky TNC 620, řezné hodnoty jsou z ní
a nemusí sedět pro váš stroj.

0  BEGIN PGM LINEAR MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20       ; polotovar, bod MIN
2  BLK FORM 0.2 X+100 Y+100 Z+0      ; bod MAX
3  TOOL CALL 1 Z S4000               ; nástroj 1, osa Z
4  L Z+250 R0 FMAX                   ; vyjetí
5  L X-10 Y-10 R0 FMAX               ; předpolohování
6  L Z-5 R0 F1000 M3                 ; na hloubku, vřeteno
7  APPR LT X+5 Y+5 LEN10 RL F300     ; tečný nájezd s korekcí
8  L Y+95                            ; bod 2
9  L X+95                            ; bod 3
10 CHF 10                            ; zkosení 10 v rohu
11 L Y+5                             ; bod 4
12 CHF 20                            ; zkosení 20
13 L X+5                             ; zpět na bod 1
14 DEP LT LEN10 F1000                ; opuštění obrysu
15 L Z+250 R0 FMAX M2                ; odjetí a konec
16 END PGM LINEAR MM

POZOR

Mezi bloky s RL a RR musí být alespoň jeden blok R0. Korekce se aktivuje na konci prvního bloku
a první pozici najede kolmo. Proto start volíš mimo obrys, aby se díl nepoškodil.

Cyklus vrtání 200

5  CYCL DEF 200 VRTÁNÍ
   Q200=2   ; bezpečná vzdálenost
   Q201=-15 ; hloubka, znaménko dává směr
   Q206=250 ; posuv do hloubky
   Q202=5   ; hloubka přísuvu
   Q210=0   ; prodleva nahoře
   Q203=-10 ; souřadnice povrchu
   Q204=20  ; 2. bezpečná vzdálenost
   Q211=0.2 ; prodleva dole
   Q395=0   ; reference hloubky
6  L X+10 Y+10 R0 FMAX M3  ; na první díru, vřeteno
7  CYCL CALL  ; provést
8  L Y+90 R0 FMAX M99  ; další díra a provést Obr. 3.5  Parametry cyklu 200 v příručce iTNC

530: Q200 až Q204, Q210 a Q211.
Vlastník obrázku: © DR. JOHANNES
HEIDENHAIN GmbH, příručka iTNC 530 Cycle
Programming, dok. 670388-25, s. 76

3  Heidenhain TNC

POZOR

Hloubka Q201 má znaménko. Záporná vrtá dolů. Q203 je skutečná výška povrchu. Polohu najeď
s R0. Pokud je hloubka 0, cyklus se neprovede.

Stejnou práci jde pro víc děr zapsat vzorem: PATTERN DEF POS1 (X+10 Y+10 Z+0) ... a pak CYCL
CALL PAT FMAX M13.

Obr. 3.6  Definice vzoru PATTERN DEF na obrazovce iTNC 530: dialog se ptá na polohu obrábění, pomocný obrázek
ukazuje jeden bod.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
61

Kapsa cyklem 251
Cyklus udělá hrubování i dokončení, je to opce 19. Nástroj se zanoří ve středu kapsy a jede ze středu
ven. Parametry vyplníš v dialogu, grafika ukazuje každý z nich. Příklad zápisu z příručky:

Obr. 3.7  Cyklus 251 v příručce iTNC 530: vlevo strany Q218 a Q219, rádius rohu Q220 a posuv Q207, vpravo druh
frézování Q351.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
143

3  Heidenhain TNC

8  CYCL DEF 251 PRAVOÚHLÁ KAPSA
   Q215=0   ; rozsah: 0 hrubovat i dokončit
   Q218=80  ; 1. strana, délka
   Q219=60  ; 2. strana, délka
   Q220=5   ; rádius rohu
   Q368=0.2 ; přídavek na stranu
   Q224=+0  ; úhel natočení
   Q367=0   ; umístění kapsy
   Q207=500 ; frézovací posuv
   Q351=+1  ; druh frézování
   Q201=-20 ; hloubka
   Q202=5   ; hloubka přísuvu
   Q369=0.1 ; přídavek na dno
   Q206=150 ; posuv přísuvu do hloubky
   Q338=5   ; přísuv načisto
   Q200=2   ; bezpečná vzdálenost
   Q203=+0  ; souřadnice povrchu
   Q204=50  ; 2. bezpečná vzdálenost
   Q370=1   ; překrytí dráhy
   Q366=1   ; zanořování: 1 po šroubovici
   Q385=500 ; posuv obrábění načisto

Pro zanoření po šroubovici (Q366=1) nebo kývavě (Q366=2) musí být v tabulce nástrojů vyplněný
úhel ANGLE, jinak systém nahlásí chybu.

Test a spuštění

Obr. 3.8  Doporučené pořadí práce.

■ V Testování programu zvol přes PGM MGT tabulku TOOL.T, když ještě není aktivní, a stiskni
RESET + START.

■ Na stroji nejdřív po blocích, s NC-START pro každý blok.

■ U cyklů s body se systém po bloku zastaví po každém bodu.

■ Přerušený program lze znovu spustit, řiď se pokyny učitele a příručkou stroje.

Obr. 3.9  Obrazovka iTNC 530 při provádění programu (Program run, full sequence): vlevo program, vpravo grafika, dole
poloha os.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 pro popisný dialog, dok. 375 738-Z0, s. 37

4 FRÉZKA: TNC A ITNC 530

Cykly iTNC 530
Přehled cyklů iTNC 530 s parametry a příklady zápisu a postup ve školním programu
Praktická MZ_2026.

Cyklus se definuje blokem CYCL DEF a spouští CYCL CALL nebo funkcí M99 v bloku s polohou.
Definice platí, dokud ji nezměníš. Před cyklem nech nástroj na bezpečné výšce a předpolohuj ho
v R0.

PAMATUJ

Texty parametrů jsou anglicky podle příručky iTNC 530 pro NC software 60642x-04. Starší verze
mají některé parametry navíc nebo méně. Nabídka závisí na opcích stroje.

Cyklus Co dělá Kdy ho použít

200 DRILLING vrtání s přísuvy běžná díra

203 UNIVERSAL DRILLING vrtání s odlomením třísky hlubší díra

232 FACE MILLING frézování roviny čelo, řádkování

251 RECTANGULAR POCKET obdélníková kapsa kapsa bez složitého tvaru

4  Cykly iTNC 530

Cyklus Co dělá Kdy ho použít

252 CIRCULAR POCKET kruhová kapsa zahloubení, kruhová kapsa

253 SLOT MILLING přímá drážka drážky, i natočené

254 CIRCULAR SLOT kruhová drážka drážka na oblouku

256, 257 obdélníkový a kruhový čep vnější tvary

14 + 20 + 22, 23, 24 kapsa podle vlastní kontury ostrůvky a složité tvary

25 CONTOUR TRAIN obrábění podél jedné kontury vnější obrys

220, 221 vzor na kružnici a v síti díry v pravidelné řadě

Společné parametry cyklů 25x
Parametr Text Význam

Q215 MACHINING OPERATION 0 hrubování a dokončení, 1 hrubování, 2
dokončení

Q368, Q369 ALLOWANCE FOR SIDE, FLOOR přídavek na bok a na dno

Q207, Q385 FEED RATE FOR MILLING,
FINISHING

posuv hrubování a dokončení

Q351 CLIMB OR UP-CUT +1 sousledně, -1 nesousledně

Q201 DEPTH hloubka, se znaménkem mínus

Q202, Q338 PLUNGING DEPTH, INFEED FOR
FINISHING

přísuv, přísuv dokončení, 0 znamená na
jednou

Q206 FEED RATE FOR PLNGNG posuv do hloubky

Q200 SET-UP CLEARANCE bezpečná vzdálenost nad plochou

Q203 SURFACE COORDINATE souřadnice povrchu

Q204 2ND SET-UP CLEARANCE druhá bezpečná vzdálenost

Q366 PLUNGE 0 kolmo, 1 šroubovice, 2 kývavě, 1 a 2
potřebují ANGLE v tabulce

4  Cykly iTNC 530

Obr. 4.1  Parametry v ose nástroje: vlevo hloubka Q201, přísuvy Q202 a Q338 a posuv Q206, vpravo Q200, Q203, Q204
a přídavky Q368 a Q369.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
144

Obdélníková kapsa: cyklus 251

8 CYCL DEF 251 RECTANGULAR POCKET
   Q215=0   ;MACHINING OPERATION
   Q218=80  ;1ST SIDE LENGTH
   Q219=60  ;2ND SIDE LENGTH
   Q220=5   ;CORNER RADIUS
   Q368=0.2 ;ALLOWANCE FOR SIDE
   Q224=+0  ;ANGLE OF ROTATION
   Q367=0   ;POCKET POSITION
   Q207=500 ;FEED RATE FOR MILLING
   Q351=+1  ;CLIMB OR UP-CUT
   Q201=-20 ;DEPTH
   Q202=5   ;PLUNGING DEPTH
   Q369=0.1 ;ALLOWANCE FOR FLOOR
   Q206=150 ;FEED RATE FOR PLNGNG
   Q338=5   ;INFEED FOR FINISHING
   Q200=2   ;SET-UP CLEARANCE
   Q203=+0  ;SURFACE COORDINATE
   Q204=50  ;2ND SET-UP CLEARANCE
   Q370=1   ;TOOL PATH OVERLAP
   Q366=1   ;PLUNGE
   Q385=500 ;FEED RATE FOR FINISHING
9 CYCL CALL POS X+50 Y+50 Z+0 FMAX M3

Obr. 4.2  Poloha nástroje při volání cyklu podle Q367: 0
střed kapsy, 1 až 4 rohy.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 143

Příklad je přímo z příručky. Při Q367=0 je poloha nástroje při volání střed kapsy. Kapsu 80 × 60
s rohy R5 a hloubkou 20 mm obrobí po 5 mm na hloubku a dokončí boky i dno.

Kruhová kapsa: cyklus 252

Zadává se průměr Q223 místo stran a rohového
rádiusu. Ostatní parametry jsou stejné jako
u 251. Nástroj před voláním stojí nad středem.

Obr. 4.3  Cyklus 252 v příručce iTNC 530: průměr Q223
a posuv Q207.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 148

Drážka: cyklus 253

CYCL DEF 253 SLOT MILLING
   Q215=+0    ;MACHINING OPERATION
   Q218=+30   ;SLOT LENGTH
   Q219=+5    ;SLOT WIDTH
   Q368=+0.1  ;ALLOWANCE FOR SIDE
   Q374=+90   ;ANGLE OF ROTATION
   Q367=+0    ;SLOT POSITION
   Q207=+60   ;FEED RATE FOR MILLING
   Q351=-1    ;CLIMB OR UP-CUT
   Q201=-6    ;DEPTH
   Q202=+0.75 ;PLUNGING DEPTH
   Q369=+0.1  ;ALLOWANCE FOR FLOOR
   Q206=+20   ;FEED RATE FOR PLNGNG
   Q338=+0    ;INFEED FOR FINISHING
   Q200=+2    ;SET-UP CLEARANCE
   Q203=-3    ;SURFACE COORDINATE
   Q204=+50   ;2ND SET-UP CLEARANCE
   Q366=+0    ;PLUNGE
   Q385=+60   ;FEED RATE FOR FINISHING
L X+90 Y+37.5 R0 FMAX
CYCL CALL

Obr. 4.4  Cyklus 253 v příručce iTNC 530: délka Q218,
šířka Q219 a natočení Q374.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 152

Svislá drážka z programu MZ 2026. Q374=+90 otočí drážku, délka Q218 je tedy ve směru Y. Když je
drážka širší než dvojnásobek průměru nástroje, řízení ji hrubuje zevnitř ven.

Kruhová drážka: cyklus 254
Zadává se průměr roztečné kružnice Q375, střed Q216 a Q217, počáteční úhel Q376 a úhlová délka
Q248. Když je šířka drážky rovná průměru nástroje, řízení dělá jen hrubování.

Obr. 4.5  Cyklus 254: vlevo šířka Q219, roztečný průměr Q375, počáteční úhel Q376 a úhlová délka Q248, vpravo
úhlový krok Q378 mezi dvěma drážkami.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
158 a 159

Díry: cyklus 200

CYCL DEF 200 DRILLING
   Q200=+2   ;SET-UP CLEARANCE
   Q201=-2.5 ;DEPTH
   Q206=+20  ;FEED RATE FOR PLNGNG
   Q202=+1   ;PLUNGING DEPTH
   Q210=+0   ;DWELL TIME AT TOP
   Q203=-2   ;SURFACE COORDINATE
   Q204=+50  ;2ND SET-UP CLEARANCE
   Q211=+0.1 ;DWELL TIME AT DEPTH
L X+76 Y+63.838 R0 FMAX M99
L X+84 Y+64.511 R0 FMAX M99

M99 po každém nájezdu spustí naposledy definovaný cyklus. Znaménko hloubky Q201 určuje směr
obrábění, proto je záporné. Přísuv Q202 se zadává kladně.

Vzor na kružnici a v síti
Cyklus Parametry Co zadáš

220 POLAR PATTERN Q216, Q217, Q244, Q245, Q246,
Q247, Q241

střed, roztečný průměr,
počáteční a koncový úhel, krok,
počet

221 CARTESIAN PATTERN Q225, Q226, Q237, Q238, Q242,
Q243, Q224

počátek, rozteče, sloupce, řádky,
natočení

Před vzorem se definuje obráběcí cyklus, například 200. Vzor ho pak sám zavolá na všech polohách.

Obr. 4.6  Vlevo vzor na kružnici, cyklus 220, vpravo vzor v síti, cyklus 221, s parametry z tabulky.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
176 a 179

Kapsa podle kontury: SL cykly
Když tvar není obdélník ani kruh, popíšeš ho v podprogramu LBL a použiješ cykly SL. První kontura
je kapsa, další jsou ostrůvky. Kapsa musí začínat mimo ostrůvek a ostrůvek uvnitř kapsy.

Obr. 4.7  Vlevo kapsa složená ze dvou kontur A a B s ostrůvky C a D, cyklus 14. Vpravo směr Q9 v cyklu 20 pro kapsu
a ostrůvek.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
187 a 193

Cyklus Účel Důležité parametry

14 CONTOUR GEOMETRY seznam kontur CONTOUR LABEL 1/2/3

20 CONTOUR DATA společná data Q1 hloubka, Q2 překrytí, Q3 přídavek bok, Q4
dno, Q5 povrch, Q6 bezpečnost, Q7 bezpečná
výška, Q9 směr

22 ROUGH-OUT hrubování Q10 přísuv, Q11, Q12 posuvy, Q18 hrubší nástroj,
Q19 kývavý posuv

23 FLOOR FINISHING dokončení dna Q11, Q12

24 SIDE FINISHING dokončení boků Q9 směr, Q10, Q11, Q12, Q14 přídavek

25 CONTOUR TRAIN podél kontury Q1, Q3, Q5, Q7, Q10, Q11, Q12, Q15

POZOR

Q9 má v cyklech jiný význam. V cyklu 20 je -1 nesousledně a +1 sousledně pro kapsu a ostrůvek.
V cyklu 24 je -1 po směru hodin a +1 proti. Cyklus 25 zná jen svou jedinou konturu, v úzkých místech
proto může poškodit jiné stěny.

Obr. 4.8  Vlevo výšky v cyklu 20: povrch Q5, bezpečnost Q6, bezpečná výška Q7 a hloubka Q1, s přísuvem Q10. Vpravo
obrábění podél kontury, cyklus 25.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
193 a 206

■ Q2: krát poloměr nástroje dává krok. Q2=1,4 u frézy Ø10 je 7 mm.

■ Q18: číslo hrubšího nástroje. Pak se dohrubuje jen to, kam se předchozí nedostal. Nástroj musí
být v tabulce.

■ Q19=0: vždy se zanořuje kolmo, jinak podle ANGLE v tabulce. Bez ANGLE přijde chyba.

■ Q6 a Q7: Q6 je vzdálenost špičky nástroje od povrchu, přírůstkově. Q7 je absolutní výška, ve
které nástroj nemůže narazit do obrobku.

Ukázka: kontura s RND

14 CYCL DEF 14.0 CONTOUR GEOMETRY
15 CYCL DEF 14.1 CONTOUR LABEL 1
16 CYCL DEF 20 CONTOUR DATA
   Q1=-7.5 ;MILLING DEPTH
   Q2=+1.4 ;TOOL PATH OVERLAP
   Q3=+0.1 ;ALLOWANCE FOR SIDE
   Q4=+0.1 ;ALLOWANCE FOR FLOOR
   Q5=+0   ;SURFACE COORDINATE
   Q6=+2   ;SET-UP CLEARANCE
   Q7=+50  ;CLEARANCE HEIGHT
   Q8=+0.2 ;ROUNDING RADIUS
   Q9=-1   ;DIRECTION OF ROTATION

17 CYCL DEF 22 ROUGH-OUT
   Q10=+0.75   ;PLUNGING DEPTH
   Q11=+20     ;FEED RATE FOR PLNGNG
   Q12=+60     ;FEED RATE FOR ROUGHING
   Q18=+0      ;COARSE ROUGHING TOOL
   Q19=+0      ;RECIPROCATION FEED RATE
   Q208=+99999 ;RETRACTION FEED RATE
18 CYCL CALL
...
26 LBL 1
27 L X+32.41 Y+28.5 RR
28 L X+25.871 Y+28.5
29 RND R2.5
30 L X+21.83 Y+40.938
31 RND R2.5
...
38 L X+32.41 Y+28.5
39 LBL 0

Postup v programu Praktická MZ_2026
Školní program v systému iTNC 530 ukazuje, jak skládat cykly za sebou. Hlavní program volá přes
CALL PGM postupně 13 samostatných programů. Uvnitř nich jsou kontury v podprogramech LBL,
které se skokem FN 9 přeskakují. Nástroj jde od největšího po nejmenší.

Program Práce Nástroj Cykly

01 ofrézování čela T40 dráhy, soustředně

02 rovina kolem trojúhelníku T20 14, 20, 22

03 vnější kontura T20 14, 25

04, 05 kapsa na Z-3 a Z-7,5 T15 14, 20, 22

06 dohrubování a dokončení T8 14, 20, 22, 23, 24

07 úzká místa T4 14, 20, 22 s Q18=8

08 pětiúhelník T4 14, 20, 22, 23, 24

09, 10 drážky T4 253

11, 12 kruhové drážky T4 254

13 díry T4 200, M99

Hlavní program z MZ 2026
Hlavní program má jen hlavičku, polotovar a volání. Konec M30 je tady správně, protože jde
o program, který nikdo další nevolá.

0 BEGIN PGM michl_mz00 MM
1 ;Programator tohoto NC: Michl Ondrej
2 ;Vyrobek: Prakticka MZ_2026, cely postup
3 ;Polotovar: 130x75x21
4 ;Zacatek: X0 Y0 levy dolni roh, Z0 o 1 pod horni plochou
5 ;Upnuti: ve sveraku, vysunuti 10
6 ;Datum: 6.11.2026
7 BLK FORM 0.1 Z X+0 Y+0 Z-20
8 BLK FORM 0.2 X+130 Y+75 Z+1

9 CALL PGM michl_mz01
10 CALL PGM michl_mz02
11 CALL PGM michl_mz03
12 CALL PGM michl_mz04
13 CALL PGM michl_mz05
14 CALL PGM michl_mz06
15 CALL PGM michl_mz07
16 CALL PGM michl_mz08
17 CALL PGM michl_mz09
18 CALL PGM michl_mz10
19 CALL PGM michl_mz11
20 CALL PGM michl_mz12
21 CALL PGM michl_mz13
22 M30
23 END PGM michl_mz00 MM

Další cykly a jejich parametry
Cyklus Parametr Význam podle příručky

203 Q212 o kolik se zmenší přísuv Q202 po každém přísuvu

203 Q213, Q256 počet odlomení třísky, vzdálenost při odlomení

232 Q389 0 a 1 meandr, 2 po řádcích

232 Q225, Q226, Q227 startovní bod ve všech třech osách

232 Q386 koncový bod v ose nástroje

232 Q218, Q219 délka a šířka plochy

232 Q357 boční bezpečná vzdálenost

256 Q424, Q425 rozměr polotovaru, větší než hotový čep

256 Q437 poloha nájezdu, 0 zprava, 1 až 4 rohy

257 Q223, Q222 hotový průměr a průměr polotovaru

257 Q376 počáteční úhel

254 Q378, Q377 úhlový krok a počet opakování drážky

Obr. 4.9  Vlevo cyklus 232 se strategií Q389=2, tedy po řádcích. Vpravo cyklus 256: hotový čep Q218 a Q219, polotovar
Q424 a Q425.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, příručka iTNC 530 Cycle Programming, dok. 670388-25, s.
272 a 163

Frézování roviny: cyklus 232

CYCL DEF 232 FACE MILLING
   Q389=2    ;STRATEGY
   Q225=+10  ;STARTING POINT 1ST AXIS
   Q226=+12  ;STARTING POINT 2ND AXIS
   Q227=+2.5 ;STARTING POINT 3RD AXIS
   Q386=-3   ;END POINT IN 3RD AXIS
   Q218=150  ;1ST SIDE LENGTH
   Q219=75   ;2ND SIDE LENGTH
   Q202=2    ;MAX. PLUNGING DEPTH
   Q369=0.5  ;ALLOWANCE FOR FLOOR
   Q370=1    ;MAX. OVERLAP
   Q207=500  ;FEED RATE FOR MILLING
   Q385=800  ;FEED RATE FOR FINISHING
   Q253=2000 ;F PRE-POSITIONING
   Q200=2    ;SET-UP CLEARANCE
   Q357=2    ;CLEARANCE TO SIDE
   Q204=2    ;2ND SET-UP CLEARANCE

Obr. 4.10  Cyklus 232 v rovině: startovní bod Q225
a Q226, délka Q218 a šířka Q219.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 273

Vrtání s odlomením třísky: cyklus 203

CYCL DEF 203 UNIVERSAL DRILLING
   Q200=2    ;SET-UP CLEARANCE
   Q201=-20  ;DEPTH
   Q206=150  ;FEED RATE FOR PLNGNG
   Q202=5    ;PLUNGING DEPTH
   Q210=0    ;DWELL TIME AT TOP
   Q203=+20  ;SURFACE COORDINATE
   Q204=50   ;2ND SET-UP CLEARANCE
   Q212=0.2  ;DECREMENT
   Q213=3    ;NR OF BREAKS
   Q205=3    ;MIN. PLUNGING DEPTH
   Q211=0.25 ;DWELL TIME AT DEPTH
   Q208=500  ;RETRACTION FEED RATE
   Q256=0.2  ;DIST FOR CHIP BRKNG
   Q395=0    ;DEPTH REFERENCE Obr. 4.11  Cyklus 203 v příručce iTNC 530: hloubka,

přísuv a prodlevy Q210 a Q211.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 85

5 FRÉZKA: TNC A ITNC 530

Řešené programy z příručky iTNC
530
Tři programy z příručky cyklů: díry, čep s kapsou a drážkami a vzory děr na kružnici.

Programy jsou z příručky cyklů. Polotovar i nástroje jsou v nich definované, takže se dají přepsat
a pustit v Testu programu. Komentáře jsou přeložené.

Vrtání čtyř děr
Polotovar 100 × 100 × 20, nástroj s poloměrem 3. První díra se spustí pomocí CYCL CALL, další tři
pomocí M99 přímo v bloku s polohou. Stačí dopsat jen osu, která se mění. V příkladu příručky je
Q203=-10, i když horní plocha polotovaru je Z+0. Ve svém programu zadej skutečný povrch.

0 BEGIN PGM C200 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 1 Z S4500
4 L Z+250 R0 FMAX            ;odjetí
5 CYCL DEF 200 DRILLING
   Q200=2   ;SET-UP CLEARANCE
   Q201=-15 ;DEPTH
   Q206=250 ;FEED RATE FOR PLNGNG
   Q202=5   ;PLUNGING DEPTH
   Q210=0   ;DWELL TIME AT TOP
   Q203=-10 ;SURFACE COORDINATE
   Q204=20  ;2ND SET-UP CLEARANCE
   Q211=0.2 ;DWELL TIME AT DEPTH
   Q395=0   ;DEPTH REFERENCE
6 L X+10 Y+10 R0 FMAX M3     ;díra 1, vřeteno
7 CYCL CALL
8 L Y+90 R0 FMAX M99         ;díra 2
9 L X+90 R0 FMAX M99         ;díra 3
10 L Y+10 R0 FMAX M99        ;díra 4
11 L Z+250 R0 FMAX M2
12 END PGM C200 MM

Obr. 5.1  Výkres k programu C200 v příručce: polotovar
100 × 100 a čtyři díry.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 101

Čep, kruhová kapsa a dvě kruhové drážky
Nástroj 1 s poloměrem 6 obrobí obdélníkový čep 90 × 80 a kruhovou kapsu Ø50. Nástroj 2
s poloměrem 3 vyfrézuje dvě kruhové drážky na roztečné kružnici Ø70. Střed je zadaný přímo v cyklu
254 pomocí Q216 a Q217, proto se nástroj před voláním nemusí v X a Y předpolohovat. Q378=180
říká, že druhá drážka začíná o 180° dál, a Q377=2 je jejich počet.

0 BEGIN PGM C210 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+6  ;hrubování a dokončení
4 TOOL DEF 2 L+0 R+3  ;drážkovací fréza
5 TOOL CALL 1 Z S3500
6 L Z+250 R0 FMAX
7 CYCL DEF 256 RECTANGULAR STUD
   Q218=90  ;1ST SIDE LENGTH
   Q424=100 ;WORKPC. BLANK SIDE 1
   Q219=80  ;2ND SIDE LENGTH
   Q425=100 ;WORKPC. BLANK SIDE 2
   Q220=0   ;CORNER RADIUS
   Q368=0   ;ALLOWANCE FOR SIDE
   Q224=0   ;ANGLE OF ROTATION
   Q367=0   ;STUD POSITION
   Q207=250 ;FEED RATE FOR MILLING

Obr. 5.2  Výkres k programu C210 v příručce: čep 90 ×
80, kapsa Ø50 a dvě drážky na roztečné kružnici Ø70.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 169

   Q351=+1  ;CLIMB OR UP-CUT
   Q201=-30 ;DEPTH
   Q202=5   ;PLUNGING DEPTH
   Q206=250 ;FEED RATE FOR PLNGNG
   Q200=2   ;SET-UP CLEARANCE
   Q203=+0  ;SURFACE COORDINATE
   Q204=20  ;2ND SET-UP CLEARANCE
   Q370=1   ;TOOL PATH OVERLAP
   Q437=1   ;APPROACH POSITION
8 CYCL CALL POS X+50 Y+50 Z+0 M3

9 CYCL DEF 252 CIRCULAR POCKET
   Q215=0   ;MACHINING OPERATION
   Q223=50  ;CIRCLE DIAMETER
   Q368=0.2 ;ALLOWANCE FOR SIDE
   Q207=500 ;FEED RATE FOR MILLING
   Q351=+1  ;CLIMB OR UP-CUT
   Q201=-30 ;DEPTH
   Q202=5   ;PLUNGING DEPTH
   Q369=0.1 ;ALLOWANCE FOR FLOOR
   Q206=150 ;FEED RATE FOR PLNGNG
   Q338=5   ;INFEED FOR FINISHING
   Q200=2   ;SET-UP CLEARANCE
   Q203=+0  ;SURFACE COORDINATE
   Q204=50  ;2ND SET-UP CLEARANCE
   Q370=1   ;TOOL PATH OVERLAP
   Q366=1   ;PLUNGE
   Q385=750 ;FEED RATE FOR FINISHING
10 CYCL CALL POS X+50 Y+50 Z+0 FMAX

11 L Z+250 R0 FMAX M6        ;výměna nástroje
12 TOOL CALL 2 Z S5000
13 CYCL DEF 254 CIRCULAR SLOT
   Q215=0   ;MACHINING OPERATION
   Q219=8   ;SLOT WIDTH
   Q368=0.2 ;ALLOWANCE FOR SIDE
   Q375=70  ;PITCH CIRCLE DIA.
   Q367=0   ;REF. SLOT POSITION
   Q216=+50 ;CENTER IN 1ST AXIS
   Q217=+50 ;CENTER IN 2ND AXIS
   Q376=+45 ;STARTING ANGLE
   Q248=90  ;ANGULAR LENGTH
   Q378=180 ;STEPPING ANGLE
   Q377=2   ;NUMBER OF OPERATIONS
   Q207=500 ;FEED RATE FOR MILLING

   Q351=+1  ;CLIMB OR UP-CUT
   Q201=-20 ;DEPTH
   Q202=5   ;PLUNGING DEPTH
   Q369=0.1 ;ALLOWANCE FOR FLOOR
   Q206=150 ;FEED RATE FOR PLNGNG
   Q338=5   ;INFEED FOR FINISHING
   Q200=2   ;SET-UP CLEARANCE
   Q203=+0  ;SURFACE COORDINATE
   Q204=50  ;2ND SET-UP CLEARANCE
   Q366=1   ;PLUNGE
   Q439=0   ;FEED RATE REFERENCE
14 CYCL CALL FMAX M3
15 L Z+250 R0 FMAX M2
16 END PGM C210 MM

Dva vzory děr na kružnici
Definice cyklu 200 stačí jedna. Cyklus 220 ho na každé pozici zavolá sám a platí v něm jeho hodnoty
Q200, Q203 a Q204. První vzor má 10 děr na celé kružnici Ø50. Druhý má 5 děr na kružnici Ø70,
začíná na 90° a další jdou po 30°.

0 BEGIN PGM PATTERN MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3
4 TOOL CALL 1 Z S3500
5 L Z+250 R0 FMAX M3
6 CYCL DEF 200 DRILLING
   Q200=2    ;SET-UP CLEARANCE
   Q201=-15  ;DEPTH
   Q206=250  ;FEED RATE FOR PLNGNG
   Q202=4    ;PLUNGING DEPTH
   Q210=0    ;DWELL TIME
   Q203=+0   ;SURFACE COORDINATE
   Q204=0    ;2ND SET-UP CLEARANCE
   Q211=0.25 ;DWELL TIME AT DEPTH
   Q395=0    ;DEPTH REFERENCE

Obr. 5.3  Výkres k programu PATTERN v příručce: vzory
děr na kružnicích R25 a R35.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN
GmbH, příručka iTNC 530 Cycle Programming, dok.
670388-25, s. 180

7 CYCL DEF 220 POLAR PATTERN
   Q216=+30  ;CENTER IN 1ST AXIS
   Q217=+70  ;CENTER IN 2ND AXIS
   Q244=50   ;PITCH CIRCLE DIA.
   Q245=+0   ;STARTING ANGLE
   Q246=+360 ;STOPPING ANGLE
   Q247=+0   ;STEPPING ANGLE
   Q241=10   ;NUMBER OF REPETITIONS
   Q200=2    ;SET-UP CLEARANCE
   Q203=+0   ;SURFACE COORDINATE
   Q204=100  ;2ND SET-UP CLEARANCE
   Q301=1    ;MOVE TO CLEARANCE
   Q365=0    ;TYPE OF TRAVERSE

8 CYCL DEF 220 POLAR PATTERN
   Q216=+90  ;CENTER IN 1ST AXIS
   Q217=+25  ;CENTER IN 2ND AXIS
   Q244=70   ;PITCH CIRCLE DIA.
   Q245=+90  ;STARTING ANGLE
   Q246=+360 ;STOPPING ANGLE
   Q247=30   ;STEPPING ANGLE
   Q241=5    ;NUMBER OF REPETITIONS
   Q200=2    ;SET-UP CLEARANCE
   Q203=+0   ;SURFACE COORDINATE
   Q204=100  ;2ND SET-UP CLEARANCE
   Q301=1    ;MOVE TO CLEARANCE
   Q365=0    ;TYPE OF TRAVERSE
9 L Z+250 R0 FMAX M2
10 END PGM PATTERN MM

PAMATUJ

V příručce je u bloku 6 hodnota Q395=0.25. Q395 volí vztah hloubky a v ostatních příkladech
příručky má hodnotu 0, proto je tu 0. Ve svém programu ho vyplň podle dialogu řízení.

6 SOUSTRUH: CNC PILOT A MANUALPLUS

Heidenhain: CNC PILOT
a MANUALplus
Řízení pro soustruhy: režimy, seřízení, měření nástroje, režim Naučit, cykly a DIN.

Heidenhain vyrábí řídicí systémy pro soustruhy MANUALplus 620 a CNC PILOT 640. Oba jsou
určené pro soustruhy a mají společnou dokumentaci. U MANUALplus je silnou stránkou práce
s cykly, kdy se dá vyrábět i bez psaní programu. Co přesně je k dispozici, závisí na systému a na
opcích stroje.

Obr. 6.1  Obsluha u soustruhu s řízením CNC PILOT a ovládací panel CNC PILOT. Školní stroj může vypadat jinak.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 32
a 3

PAMATUJ

Heidenhain TNC jsou řízení hlavně pro frézky, jiný systém s jiným jazykem. Tato kapitola platí pro
soustruhy s MANUALplus a CNC PILOT. Názvy tlačítek se mohou lišit podle verze softwaru, kterou
má váš stroj, a podle výrobce stroje.

Provozní režimy

Obr. 6.2  Čtyři hlavní provozní režimy a jejich podřízené režimy. Přepínají se klávesami provozních režimů.

Obrazovka má nahoře řádek režimů, pod ním strojní indikaci (polohy os, posuv, otáčky, nástroj).
Funkce se volí softklávesami pod obrazovkou a nabídkou s devíti poli.

Obr. 6.3  Obrazovka CNC PILOT 640 v režimu Stroj: nahoře řádek provozních režimů, pod ním strojní indikace, dole
softklávesy a okno nabídky s devíti poli.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 49

Zapnutí a přejetí referencí
1. Zapni stroj. Systém provede testy a přejde do režimu Stroj.

2. Zkontroluj případná hlášení. Systém předpokládá, že je upnutý naposledy použitý nástroj. Když
ne, zadej ho.

3. Podle typu snímačů proveď přejetí referencí: softklávesa Z-reference, X-reference nebo
Všechny a potom Start cyklu.

4. Stroj se snímači EnDat reference nepotřebuje, s distančně kódovanými stačí krátké přejetí, se
standardními osy jedou na známá místa.

Obr. 6.4  Softklávesy přejetí referencí na CNC PILOT 640: Z, X, Y a všechno.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 72

Ruční provoz: manuální práce na CNC stroji
Po zapnutí a přejetí referencí je systém v Ručním provozu. Nástrojem pohybuješ ručními kolečky
nebo směrovými tlačítky, podobně jako na klasickém soustruhu. Můžeš přitom použít i cykly, tomu
se říká poloautomatický provoz. Dráhy ani cykly se tady neukládají.

Dialog TSF
Pro ruční práci zadáváš nástroj, otáčky a posuv v dialogu TSF:

Písmeno Význam Poznámka

T číslo nástroje T0 je bez nástroje, cykly s ním nelze použít.

S otáčky vřetena Hlídej maximální otáčky. Vřeteno se zapíná tlačítky na panelu.

F posuv Při točícím se vřetenu v mm/ot, při stojícím v mm/min.

Obr. 6.5  Zadávací okno Nastavení T, S, F na obrazovce CNC PILOT 640.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 74

POZOR

Před ručním obráběním musí být zadaný nulový bod obrobku a strojní data. Mysli na limit otáček.

Seřízení: nulový bod obrobku
1. V Ručním provozu zvol Seřizování a potom Nastavení hodnot os.

2. Naškrábni nebo zarovnej čelo nástrojem se změřenými rozměry.

3. Tuto polohu definuj jako Nulový bod obrobku Z. Zadej vzdálenost nástroje od nuly, když stojí
přesně na ní, zadáš 0.

4. Systém spočítá a uloží posunutí. Vzdálenost M až W ukazuje jako XN a ZN.

Obr. 6.6  Okno Nastavení hodnoty osy. Pomocný obrázek ukazuje vzdálenost mezi body M a W jako XN a ZN.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 89

Ve stejné nabídce se nastavuje i bezpečnostní pásmo a bod výměny nástroje. Bod výměny je
důležitý, cyklus Najetí do bodu výměny a příkaz G14 ho používají.

Měření nástroje naškrábnutím
Heidenhain podporuje měření naškrábnutím, dotykovou sondou a měřicí optikou. Naškrábnutí je
k dispozici vždy:

1. Zapiš měřený nástroj do tabulky nástrojů, případně v režimu Správa nástrojů.

2. Zarovnej čelo změřeným nástrojem a polohu nastav jako nulový bod obrobku.

3. V dialogu TSF vyměň nástroj za měřený a aktivuj Měření nástroje.

4. Naškrábni čelo a jako souřadnici Z měřeného bodu zadej 0, ulož.

5. Osoustruž měřený průměr, změř ho a zadej jako souřadnici X, ulož.

6. U soustružnického nástroje zadej rádius břitu a převezmi ho do tabulky.

Obr. 6.7  Pomocný obrázek k měření nástroje naškrábnutím: body M a W, rozměry Z a ØX.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 98

POZOR

Korekční hodnoty se při měření nástroje smažou. Zkontroluj, že jsi je opravdu nechtěl zachovat.

Režim Naučit (Teach-in)
Režim Naučit je největší výhoda systému pro kusovou práci. Díl obrábíš krok za krokem pomocí
cyklů, ale systém si je zároveň ukládá jako program. Není potřeba psát G kódy. Postup pro každý
cyklus je:

1. Zvol cyklus z nabídky a zadej parametry včetně nástroje, otáček a posuvu. V Naučit zadáváš
i bod startu.

2. Cyklus graficky překontroluj.

3. Proveď ho na stroji.

4. Ulož ho. Cykly tvoří postupně program, který pak spustíš v režimu Chod programů.

Obr. 6.8  Vlevo zadávání cyklu Podélný řez s pomocným obrázkem, vpravo režim Naučit se seznamem uložených cyklů.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 192
a 104

POZOR

V režimu Naučit najede systém do bodu startu nejkratší cestou, diagonálně rychloposuvem. Když by
mohl narazit, přidej před něj cyklus Polohování rychloposuvem.

Nabídka cyklů
Skupina Co obsahuje

Polotovary standardní a ICP polotovar, důležitý pro simulaci

Samostatné řezy polohování rychloposuvem, najetí do bodu výměny, přímka axiálně, radiálně a pod
úhlem, kruh, zkosení, zaoblení, M-funkce

Úběrové cykly hrubování a dokončování axiálně a radiálně, se zanořováním, ICP podle libovolného
obrysu

Zápichové cykly zapichování a odlehčovací zápichy, například DIN 76 za závitem

Závitové cykly válcový, kuželový a API závit, dořezávání

Vrtací cykly vrtání, hluboké vrtání, řezání závitů v díře

Frézovací cykly jen se strojem s poháněnými nástroji

ICP editor pro libovolný obrys z přímek a oblouků, používá se v ICP cyklech

Obr. 6.9  Nabídky cyklů v režimu Naučit: Jednoduchý řez a nabídka úběrových cyklů.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 130
a 144

Na pomocných obrázcích cyklů platí: čárkovaná čára je dráha rychloposuvem, plná pracovním
posuvem.

Chod programu
V režimu Chod programů se program spouští a nemění. Při nahrání systém porovná nástroje
v programu s osazením revolverové hlavy a nesoulad ohlásí.

Obr. 6.10  Vlevo režim Běh programu s nahraným programem, vpravo hlášení, že seznam nástrojů v programu
nesouhlasí s osazením revolverové hlavy.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 105
a 106

POZOR

Seznam nástrojů z programu převezmi jen tehdy, když opravdu odpovídá tomu, co je v revolveru.
Jinak hrozí kolize.

■ Před spuštěním použij grafickou simulaci.

■ Pro první kus použij provádění po blocích, potom plynulé provádění.

■ Přerušené obrábění lze navázat hledáním bloku startu.

Programování DIN ve smart.Turn
V režimu smart.Turn se píšou NC programy (přípona .nc). Program má pevné části. Obrys dílu se
popíše jednou a cykly na něj odkazují. Následující příklad vychází z prvních kroků v příručce
Heidenhain, díl je z hliníku:

Obr. 6.11  Editor smart.Turn na CNC PILOT 640 s otevřenou nabídkou G funkcí nad NC programem.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 51

Obr. 6.12  Polotovar Ø60 x 60, přídavek 1 mm na čele. Obrys je popsaný po blocích.

HLAVICKA PROGR.
#MATERIAL    Hliník
#JEDNOTKA    METRICKÁ
OTOCNA HLAVA
T1 ID"..."                      ; nástroje z tabulky
T2 ID"..."
POLOTOVAR
N1 G20 X60 Z60 K1               ; polotovar: Ø60, délka 60, 1 mm na čele
DOKONCENA SOUC.
N2 G0 X0 Z0                     ; začátek obrysu
N3 G1 X20
N4 G1 Z-10
N5 G1 X40 Z-20
N6 G1 Z-30
N7 G1 X50
N8 G1 Z-40
N9 G1 X0
OBRABENI
N10 G26 S3000                   ; omezení otáček, hodnota podle stroje
N11 G14 Q0                      ; najetí do bodu výměny
N12 T1
N13 G95 F0.4                    ; posuv mm/ot
N14 G96 S220                    ; konstantní řezná rychlost m/min
N15 M4                          ; vřeteno
N16 G0 X62 Z2                   ; najetí před polotovar
N17 G47 P2                      ; bezpečná vzdálenost 2 mm
N18 G820 NS3 NE3 P5 K0.2        ; čelní hrubování bloku 3
N19 M8
N20 G810 NS4 NE8 P5 I0.5 K0.2   ; podélné hrubování bloků 4 až 8
N21 M9

N22 G14 Q0
N23 T2
N24 G95 F0.25
N25 G96 S240
N26 G0 X62 Z2
N27 G890 NS3 NE8                ; dokončení obrysu
N28 G14 Q0
N29 M30
KONEC

PAMATUJ

Skutečný zápis vzniká v dialozích editoru, tlačítko INS vloží nový blok, přes G zadáš číslo funkce.
Příklad ukazuje logiku, ne přesný výpis obrazovky. Řezné hodnoty jsou pro hliník z příručky a na ocel
se nehodí.

Úrovně programování
Způsob Co to znamená Pro koho

Naučit (Teach-in) cykly z nabídky, formuláře, bez psaní začátek, kusová práce

DIN přímky, oblouky a jednoduché cykly základ programování

DIN PLUS polotovar a obrys z editoru ICP, cykly odkazující na
obrys

složitější díly

smart.Turn (opce) obrábění pomocí Units, tedy formulářů obráběcích
kroků

běžná výroba, pokud
je opce

TURN PLUS (opce) pracovní plán a obrábění se vytvoří automaticky pokročilí, pokud je
opce

Simulace a nástroje
■ Simulace se spouští z programování. Kontroluje dráhy a tvar a počítá časy.

■ Správa nástrojů obsahuje editor nástrojů s rozměry a editor technologie s řeznými daty. Cyklus si
z ní může převzít návrh hodnot.

■ DataPilot je program Heidenhain pro PC, na kterém se dá obsluha a programování procvičovat
mimo stroj, pokud ho škola má.

Obr. 6.13  Vlevo grafická simulace, vpravo seznam nástrojů v editoru nástrojů.
Vlastník obrázku: © DR. JOHANNES HEIDENHAIN GmbH, Příručka uživatele CNC PILOT 640, dok. 1079662-C1, s. 470
a 486

7 PRVNÍ BĚH A PROCVIČENÍ

Bezpečný první běh a chyby
Co zkontrolovat před prvním spuštěním a jak poznat nejčastější chyby.

Před prvním spuštěním
1. Program zkontroluj v Testu programu s grafikou a polotovarem z BLK FORM.

2. Ověř v tabulce nástrojů délku L a rádius R každého volaného nástroje.

3. Pro hrubší nástroj v Q18 ověř, že je v tabulce a má správné údaje.

4. Nastav vztažný bod obrobku a ulož ho.

5. Po blocích a s nízkým override pusť první průběh.

6. Po první operaci změř díl a zadej případnou opravu do DR nebo DL.

Časté chyby
Chyba Příčina Co udělat

Kolize v rychloposuvu chybí zvednutí před přejezdem Z nejdřív nahoru, přejezd v R0

Nástroj se zanoří moc
hluboko

kladná hloubka nebo chybná Q203 Q201 záporné, ověřit povrch

Chyba Příčina Co udělat

Chyba u cyklu 22 nebo 253 chybí ANGLE nebo LCUTS doplnit do TOOL.T, nebo Q19=0,
Q366=0

Kontura se neuzavře chybný koncový bod poslední bod = první bod

Kapsa je větší špatný rádius v tabulce ověřit R a DR

Program končí dřív M30 ve volaném programu odstranit M30, použít skok FN 9

Dotyk se stěnou RL a RR zaměněné kolem dílu RL, v kapse RR při
stejném směru

8 PRVNÍ BĚH A PROCVIČENÍ

Kontrolní otázky a řešení
Otázky s řešením, slovníček a zdroje, ze kterých manuál vychází.

Otázky
1. Co je popisný dialog?

2. Jaké řady Heidenhain jsou pro frézky a jaké pro soustruhy?

3. Co definuje BLK FORM?

4. Co znamená L X+10 Y+20 RL F200?

5. Čím se liší CR a C?

6. Co znamená R- u CR?

7. Co udělá RND R5?

8. Kdy se použije RL a kdy RR kolem dílu?

9. Co znamená DR-?

10. Co udělá CALL LBL 1 REP 3?

11. Proč nesmí být M30 v programu volaném přes CALL PGM?

12. Jak přeskočíš podprogramy v takovém programu?

13. Co znamená Q201?

14. Co nastavíš pro drážku natočenou o 90°?

15. Co se stane, když je šířka drážky v cyklu 254 stejná jako nástroj?

8  Kontrolní otázky a řešení

Odborný výcvik  ·  Manuál Heidenhain 42

8



16. Jaký je rozdíl mezi Q9 v cyklu 20 a v cyklu 24?

17. K čemu slouží Q18?

18. Co dělá Q19=0?

19. Kolik je krok při Q2=1,4 a fréze Ø8?

20. Proč se cyklem 25 nedělají ostrůvky v úzkém místě?

21. Jak spustíš jeden cyklus na více polohách?

22. Co je BLK FORM a proč nestačí pro kontrolu upnutí?

23. Čím se liší Q424 od Q218 u cyklu 256?

24. Co nastavíš v cyklu 232 pro řádkování s odjezdem?

25. Kolik drážek udělá cyklus 254 s Q377=2 a Q378=180 a kde začíná druhá?

26. Proč lze v programu C200 psát jen L Y+90 R0 FMAX M99?

27. Proč v programu C210 není před cyklem 254 nájezd v X a Y?

28. Kolik děr obrobí Q241=10 v cyklu 220 a kde je definice vrtání?

29. Proč je M30 v michl_mz00 správně, ale v michl_mz05 by vadilo?

30. Co dělá TOOL DEF v programu?

Řešení
1. Program se píše odpovědí na otázky řízení: Klartext.

2. Frézky iTNC 530, TNC 620, TNC 640. Soustruhy MANUALplus 620, CNC PILOT 640.

3. Polotovar pro grafickou kontrolu.

4. Přímka do X10 Y20, nástroj vlevo od dráhy, posuv 200.

5. CR zadává poloměr, C oblouk kolem středu CC.

6. Oblouk větší než 180°.

7. Zaoblí roh mezi dvěma drahami poloměrem 5.

8. RL je sousledné po směru hodin, RR nesousledné.

9. Oblouk ve směru hodinových ručiček.

10. Zopakuje část programu třikrát, celkem proběhne čtyřikrát.

11. Program by se ukončil, nevrátil by se do hlavního.

12. Skokem FN 9: IF +0 EQU +0 GOTO LBL 99.

13. Hloubku, záporná hodnota určuje směr.

14. Q374=+90.

15. Dělá jen hrubování.

16. V cyklu 20 je to sousledné nebo nesousledné, v cyklu 24 přímo směr pohybu.

17. Číslo hrubšího nástroje, dohrubuje se jen zbytek.

18. Vždy se zanořuje kolmo.

19. 1,4 · 4 = 5,6 mm.

20. Zná jen svou konturu a mohl by poškodit protější stěnu.

21. Definuje cyklus a v blocích s polohou použije M99.

22. Popisuje jen polotovar. Upínky ani svěrák v něm nejsou, proto se kontroluje i upnutí.

23. Q424 je rozměr polotovaru, Q218 hotového čepu. Polotovar musí být větší.

24. Q389=2.

25. Dvě. Druhá začíná o 180° dál od první.

26. X zůstává z předchozího bloku a M99 zavolá naposledy definovaný cyklus.

27. Střed je v cyklu přímo v Q216 a Q217.

28. Deset. Vrtání je v cyklu 200 definovaném před vzorem.

29. Hlavní program nikdo nevolá. Volaný program musí vrátit řízení, proto v něm M2 a M30 nesmí
být.

30. Definuje délku a rádius nástroje přímo v programu.

Slovníček
Pojem Vysvětlení

Klartext popisný dialog Heidenhain

LBL návěstí začátku podprogramu nebo části

Q parametr proměnná, u cyklů označuje vstup

SL cykly cykly pro kapsy podle vlastní kontury

TOOL.T tabulka nástrojů

Vztažný bod nula obrobku

Zdroje
■ HEIDENHAIN: iTNC 530, Cycle Programming, dokument 670388-25, NC software 60642x-04,

content.heidenhain.de.

■ HEIDENHAIN: iTNC 530, příručka pro popisný dialog, čínské vydání 375 738-Z0 z 4/2005 pro
starší software, použité jen pro obrázky bez čínského textu, content.heidenhain.de.

■ HEIDENHAIN: TNC 620, Příručka pro uživatele, programování s popisným dialogem, dokument
1096883-C2, a příručka cyklů, dokument 1096886-C2.

■ HEIDENHAIN: CNC PILOT 640 a MANUALplus 620, příručky pro uživatele, dokumenty
1079662-C1 a 1118606-C7.

■ Programy Praktická MZ_2026 podle školních výkresů. Ani ony, ani příklady z příručky nebyly pro
tento manuál spuštěné na stroji nebo v Testu programu.

■ Logo HEIDENHAIN: © DR. JOHANNES HEIDENHAIN GmbH, zadní strana příručky iTNC 530
Cycle Programming, dokument 670388-25, s. 533.

■ Řídicí panely, obrazovky a obrázky cyklů: © DR. JOHANNES HEIDENHAIN GmbH, příručky iTNC
530 Cycle Programming (dokument 670388-25), iTNC 530 pro popisný dialog (dokument 375
738-Z0, čínské vydání, použité jen obrázky bez čínského textu) a CNC PILOT 640 (dokument
1079662-C1). Dokument a strana jsou uvedené pod každým obrázkem. Použito pro výuku.
Schémata bez uvedeného vlastníka jsou vlastní zpracování.

Odkud jsou obrázky

Kde v manuálu iTNC 530 Cycle Programming,
dok. 670388-25

iTNC 530 popisný dialog,
dok. 375 738-Z0

CNC PILOT 640, dok.
1079662-C1

Obálka
a záhlaví

panel s. 1, logo s. 533 panel s. 1

Kapitola 1 s. 1, 50 s. 1

Kapitola 2 s. 54 s. 162, 187, 190, 194,
195, 197, 198

Kapitola 3 s. 61, 76, 143 s. 37, 39, 40, 41, 48

Kapitola 4 s. 85, 143, 144, 148, 152, 158,
159, 163, 176, 179, 187, 193,
206, 272, 273

Kapitola 5 s. 101, 169, 180

Kapitola 6 s. 3, 32, 49, 51, 72, 74,
89, 98, 104, 105, 106,
130, 144, 192, 470,
486

PAMATUJ

Manuál nenahrazuje příručku ke stroji ani školení. Možnosti konkrétního stroje a opce ověř s učitelem
odborného výcviku.

